5 FUNCTIONS

Sections

- 5.1 Maple Functions
- 5.2 Polynomials
- 5.3 Trigonometry
- 5.4 Powers and Roots
- 5.5 Miscellaneous
- 5.6 Functional Notation
- 5.7 Application (TBD)
- Summary
- Key Terms
- Problems

Objectives

After reading this chapter, you should be able to:

- Identify the syntax of Maple functions
- Distinguish between function inputs and output
- Find functions from the Maple libraries
- Simulate functions with assignments
- Use functions for polynomials, trigonometry, logarithms, and more
- Develop customized Maple functions using functional notation

Professional Success: Understanding Functions

Introductory courses in engineering and science provide tools for later study. However, mathematical tools seem to fixate on apparently unimportant theories and concepts. Many students often complain, "Why are we learning *this*? How on earth will *this* ever help us?" Knowing how underlying theories reflect physical principles helps to motivate study. Consider, for instance,a why we really need *functions*. Functions provide realistic physical models, as demonstrated in the following example.



Figure 5-1 Experiments

Consider the experiment shown in Figure 5-1. A weight w hangs from an unstretched spring and is slowly released. The final displacement u is then measured and plotted. Different weights yield different displacements, as shown in Figure 5-2. Numerous experiments eventually develop a pattern, which is a line drawn through the individual points.



Figure 5-2 Measured Results

Because w is chosen for arbitrary values, w is called an *independent variable* and is plotted on the horizontal axis. Independent variables are not measured during the experiment. The displacement u depends on the choice of w. Experimentally measured variables are called *dependent variables* and are plotted on the vertical axis.



Figure 5-3 Ideal Results

Each weight causes a unique displacement. Thus, u is a *function* of w. In mathematical terms, $u = f(w)$. From Figure 5-3, the relationship appears to be linear. So, you may express the slope as $\frac{1}{k}$, where k is the stiffness of the spring. Using k, you may express u as $u = \frac{w}{k}$ or just $w = ku$.



Figure 5-4 Fake Results

Why choose a function? Consider the fake model shown in Figure 5-4. This model predicts that one weight can yield *different* displacements! Remember the "vertical-line test" for functions? This fake relationship fails that test.

This example shows that rules for defining functions are crucial for building accurate and realistic models. Now, consider other mathematical theories. Very often mathematical definitions and theories grow from desires to solve certain problems. Though the reasons might seem obscure, have faith that many of your early studies truly have importance!

5.1 Maple Functions

One of Maple's most common expressions is the *function*. You have already used some of Maple's wealth of predefined functions to assist your work. This section reviews further aspects of Maple functions. Inspect `?inifcn` and `?index[function]` for a multitude of Maple library functions. Also, consult Appendix C for methods on finding, loading, and viewing functions.

5.1.1 Terminology

A ***Maple function*** has the general syntax `expr(exprseq)`. Usually,

- `expr` will be a name which labels the function.
- `exprseq` will be an expression sequence `expr1,expr2,`....

Each ***expr*** inside ***exprseq*** is called an ***argument*** or ***parameter***. Consult **?function** and **?type[function]** for further discussion.

5.1.2 Input/Output

Arguments and parameters serve as function *inputs*, which are quantities upon the function acts. Some functions, like **plot(x^2,x=1..2)**, require multiple arguments, whereas functions like **anames()** can accept zero arguments. Though not required, many functions produce *output* based on the evaluation of the input. For instance, entering **sqrt(4)** produces the output of 2 from the input 4. You may assign the result of the function's evaluation to a variable:

Step 95: Output from a Function

```
> Result := sqrt(4);
```
Evaluate $\sqrt{4}$ and store the result in Result.

$$Result := 2$$

Maple reports the evaluation and assignment.

Sometimes programmers refer to evaluating a function as *calling a function*. You may think of the inputs as messages you pass when making the "call." Then, the function returns your "call" with an output value. In Step 95, the output was then assigned to a name for later use.

5.1.3 Nested Functions

Functions are yet another Maple type that form expressions. You can enter functions into one another as ***nested functions***, which are functions that employ functions as arguments. Why bother with nesting? Well, you can either enter expressions the long way, as in Step 96:

Step 96: Tedious Method for Many Functions

```
> A:=Pi/6: B:=sin(A): sqrt(B);
```
Solve for the square root of the sine of Pi divided by 6.

$$\frac{1}{2}\sqrt{2}$$

*Maple evaluated **sqrt(sin(Pi/6))**.*

To save time typing, use the notion that functions are expressions. Since, each input to a function is an expression, you may input another function, as in ***func1(func2(expr))***. The result of ***func2(expr)*** becomes the input for ***func1***. For example, repeat Step 96 by nesting your functions:

Step 97: Nested Functions

```
> sqrt( sin(Pi/6) );
```
*Enter **sin(Pi/6)** as an argument of **sqrt**.*

$$\frac{1}{2}\sqrt{2}$$

You get the same answer with less work!

Maple permits nesting of functions to many levels, as in ***func(func(func(...)))***. You do not need to insert extra spaces as in Step 97, though doing so makes your code easier to read. You should also ensure that your parentheses "match."

5.1.4 Operators

You may combine and manipulate functions as you would with other expressions. In general, operators should be entered *after* the right parenthesis ("`)`") that terminates a function. For instance, to evaluate $\sin^2(x)$ (the square of the sine of x), you would enter `sin(x)^2`, but *not* "`sin^2(x)`" or "`sin(x^2)`"!

Step 98: Functions and Powers

```
> sin(Pi/4)^2;
```

Determine $\sin^2\left(\frac{\pi}{4}\right) = \left(\sin\frac{\pi}{4}\right)^2$.

$$\frac{1}{2}$$

$\sin\left(\frac{\pi}{4}\right) = \frac{1}{2}\sqrt{2}$. *Thus,* $\left(\frac{1}{2}\sqrt{2}\right)^2 = \left(\frac{1}{4}\right)(2) = \frac{1}{2}$.

Remember that Maple performs full evaluation and automatic simplification on expressions composed of functions.

5.1.5 Assignments

You may use assignments to simulate functions by entering the statement ***`name:=expr`***. Maple then replaces each instance of ***`name`*** with ***`expr`***:

Step 99: Simulated Function Assignment

```
> y := 2*x+1;
```

Assign to the variable y *the expression* $2x+1$.

$$y := 2x + 1$$

So far, so good...

```
> x:=1: y;
```

Assign to the variable x *the value* 1.

$$3$$

Maple replaced x *with* 1 *inside* $2x+1$, *which produces* 3.

However, you must never enter $f(x) = mx + b$ as "`f(x):=m*x+b`" to express $f(x)$:

Step 100: Incorrect Function Assignment

```
> x:='x':
```

Unassign x.

```
> f(x) := 2*x+1;
```

Try to assign a function.

$$f(x) := 2x + 1$$

The function appears OK but is really incorrect!

```
> f(1);
```

Assign to the variable x *the value* 1.

$$\mathrm{f}(1)$$

Maple does not evaluate $f(x)$ *because you used the incorrect notation.*

Maple does permit valid functional notation $f(x)$ syntax, which is eexplained in Section 5.6.

Practice!

1. What are the inputs for plot(x,x=0..10,title="hello!")? Does this function have an output expression?

2. Enter the expression $\sin^2\theta + \cos^2\theta$ using Maple Notation.

3. Reduce the expression in the previous function with a call to the **simplify** function.

4. Simulate the equation $f(t) = \sin t$ with the name y and an assignment.

5.2 Polynomials

Polynomials are incredibly common functions because they help approximate intricate models. This section demonstrates many functions for operating on and manipulating polynomials. For an extensive overview of all of these functions, consult Mathematics...Algebra...Polynomials... in the Help Browser.

5.2.1 Definition

Polynomials contain sums of *terms* with *integer* exponents:

$$(term)^0 + (term)^1 + (term)^2 + \ldots + (term)^n \tag{5-1}$$

where each *term* represents virtually any name or constant. Not all terms and powers must be present. For example, both $x + 1$ and $x^2 + y + 30$ are polynomials. If you are unsure, you may confirm whether or not the expression is a polynomial:

Step 101: Check Polynomial Type

```
> x:='x':y:='y':
> type(x^2+y+30,polynom);
                              true
```

Clear x and y assignments.

Test whether or not x^2+y+30 is a polynomial.

x^2+y+30 is indeed a polynomial.

For more information on polynomial types, consult **?polynom**, **?content**, **?ratpoly**, **?type[polynom]**, and **?type[monomial]**.

5.2.2 Arithmetic

You may use Maple operators and functions on polynomials. Beware that sometimes Maple will not simplify the resulting expression! To demonstrate, try the following steps:

Step 102: Assign Polynomials

```
> P1 := x^2+3*x+2;
```

Assign to $P1$ a polynomial x^2+3x+2.

$$P1 := x^2 + 3x + 2$$ *Did you remember your operators* ***** *and* **+**?

```
> P2 := x+4;
```
Assign to $P2$ a polynomial.

$$P2 := x + 4$$ *Did you remember your operators* ***** *and* **+**?

Now, perform arithmetic operations:

Step 103: Polynomial Addition

```
> P1+P2;
```
Add two polynomials.

$$x^2 + 4x + 6$$ *Maple automatically simplifies common terms.*

Maple was able to add common terms. But, what about multiplication?

Step 104: Polynomial Multiplication

```
> P1*P2;
```
Multiply two polynomials.

$$(x^2 + 3x + 2)(x + 4)$$ *Chapter 7 discusses how to simplify these results.*

5.2.3 Expanding

Are you wondering why Maple did not produce $x^3 + 7x^2 + 14x + 8$ Step 104? Maple does not prefer to multiply all terms because you might need automatic simplification for division operation in further evaluations. (See Practice Problems, below.) If you wish to force the multiplication, enter **`expand(P1*P2)`**:

Step 105: Expand Polynomials

```
> expand(P1*P2);
```
expand *performs all possible multiplication and addition operations.*

$$x^3 + 7x^2 + 14x + 8$$ $(x^2+3x+2)(x+4) = x^3+7x^2+14x+8$

According to **`?expand`**, **`expand(`*`expr`*`)`** distributes products over sums. The next chapter demonstrates furthers uses of **`expand`**.

5.2.4 Factoring

You may "reverse" the effects of expand by using **`factor(`*`polynom`*`)`**. The ***factors*** of a polynomial are the smallest divisible polynomials whose product yields the polynomial. For the polynomials *P1* and *P2*, try to find their factors:

Step 106: Factor Polynomials

```
> factor(P1); factor(P2);
```
Factor $P1$ and $P2$.

$$(x + 2)(x + 1)$$ $(x+2)(x+1) = x^2+3x+2 = P1$

$$(x + 4)$$ $(x+4)$ *is the only factor of* $x+4$.

Maple simply returns the polynomial as the sole factor when no factorization is possible. For related functions and more information, consult **?factor**, **?ifactor**, and **?roots**.

Practice!

5. Determine whether or not the following statements produce a polynomial:

```
> a:='a': x:=sin(a): poly:=x^2+2;
```

6. Unassign x. Now, store $x^2 - 1$ in *A*. Store $x^2 + 3x + 2$ in *B*.

7. Evaluate $A^2 + AB$. Assign to $C1$ the result.

8. Evaluate $\frac{A}{B}$. Assign to the result $C2$.

9. Enter **expand** and/or **factor** to simplify $C1$ and $C2$.

5.2.5 Division

Given two expressions *a* and *b*, you can divide *a* by *b* by specifying $a \div b$ or *a*/*b*. The division of a by b produces a *quotient* q and a *remainder* r such that

$$a = bq + r. \tag{5-2}$$

For instance, $5 \div 3$ yields $a = 5$, $b = 3$, $q = 1$, and $r = 2$, where $5 = 3(1) + 2$. In general, to divide ***a*** by ***b***, you may enter either **quo(*a*,*b*,*term*,'r')** or **rem(*a*,*b*,*term*,'q')**. You may supply **'r'** or **'q'** as arguments to **quo** or **rem**, respectively, to automatically assign a remainder ***r*** or quotient ***q*** value:

Step 107: Divide Polynomials

```
> q := quo(P1,P2,x,'r'):
> q, r;
```

Divide a by b such that $a = bq + r$.

Report the quotient q and remainder r.

$$x - 1, 6$$

$(x+4)(x-1)+6 = x^2+4x-x-4+6 = x^2+3x+2$

For more information, consult **?rem** or **?quo**, **?divide**, **?mod**, **?gcd**, **?lcm**, **?evala**, and **?irem** or **?iquo**.

5.2.6 Root Finding

Factorable polynomials have ***roots*** that equate polynomials to zero when the roots are substituted back into the polynomial. For instance, $x^2 + 3x + 2$ factors into $(x + 2)(x + 1)$ with roots

$x = -2$ and $x = -1$, which both cause $x^2 + 3x + 2$ to become zero. Other polynomials have repeated factors. In the example, $x^2 + 2x + 1 = (x+1)(x+1)$, which has the root -1 that appears twice. Thus, the root -1 has a *multiplicity* of 2.

Maple reports polynomial roots as a list of pairs in the form $[[r_1, m_1], [r_2, m_2], \ldots, [r_n, m_n]]$. Each $[r_i, m_i]$ pair is the ith root r with multiplicity m. For instance, find the roots of the polynomial $P1 = x^2 + 3x + 2$:

Step 108: Polynomial Roots

```
> x:='x':                                         Unassign x.
> roots(x^2+3*x+2);                               Find the roots of P1.
                  [[-2, 1], [-1, 1]]              P1 has two roots, -2 and -1.
                                    -2 factors P1 only once. -1 factors P1 only once.
```

Compared with the results of **factor(x^2+3*x+2)**, you can verify that **roots** found each root of $x^2 + 3x + 2$. For more information, see **?roots**, **?root**, and **?realroot**.

Practice!

10. Evaluate the quotient q and remainder r in Step 107 with **rem**.

11. Confirm that your quotient and remainder in the above problem are valid. Hint: Use **expand**.

12. What are the roots of $x^3 - 3x - 2$? Do any roots repeat? If so, how many times?

5.3 Trigonometry

Many equations rely on trigonometry to transform physical models into different coordinate systems. After all, nature knows no axes! Trigonometry helps model a variety of problems throughout all branches of engineering and science. This section introduces basic trigonometric functions in Maple.

5.3.1 Warning! Use Radians for Angles

Many programs, including Maple, require angles to be entered in terms of ***radians***. Use the conversion

$$\frac{radians}{2\pi} = \frac{degrees}{360°} \tag{5-3}$$

or **convert(*angle*,radians)**:

Step 109: Angle Conversion to Radians

```
> convert(45*degrees,radians);
```
Enter $45°$ *as* **45*degrees**.

$$\frac{1}{4}\pi$$
Maple evaluates radians in terms of π *when possible.*

Remember to always specify π as **Pi** when entering angles with radians! Consult **?convert[degrees]** and **?convert[radians]** for more information.

5.3.2 Trigonometric Functions

Table 5-1 summarizes common trigonometric functions. The following example demonstrates that Maple sometimes uses automatic simplification with trigonometric functions. Note the use of radians:

Step 110: Automatic Simplification and Trig

```
> sin(0),sin(Pi/2),sin(Pi);
```
Find $\sin(0), \sin\left(\frac{\pi}{2}\right), \sin(\pi)$.

$$0, 1, 0$$
Maple used automatic simplification to find the answers.

Consult **?trig** for a full listing that includes hyperbolic functions. Inverse trigonometric functions are described in **?invtrig**.

Practice!

13. Convert $120°$ to radians. Concert the result back to degrees.
14. Find the secant of $30°$.
15. Find the tangent of $\frac{\pi}{2}$.
16. Assume the sine of an angle is 0.35. What is the angle in degrees?

5.4 Powers and Roots

This section reviews functions that are associated with powers and roots.

5.4.1 Exponentiation

Recall that the ***exponentiation*** operators **^** and ****** operators raise an expression to a power. In the following example, try entering 123×10^{-2} without using a **e** or **E**:

Step 111: Exponentiation

```
> 123.*10^(-2);
```
Simulate scientific notation.

$$1.230000000$$

Yes, entering **123.0E-2**. *would be quicker.*

See also

- **?arithop** and **?type[arithop]** for **^** and ******
- **?float** for scientific notation with **e** and **E**

5.4.2 Roots

You have already used **sqrt(x)** for $\sqrt{x}$. In general, you can also find the nth ***root*** of x with exponentiation and fractional powers:

$$\sqrt[n]{x} = x^{\frac{1}{n}}. \tag{5-4}$$

For instance, try finding the cube root of 8:

Step 112: Roots

```
> 8^(1/3);
```
Find the cube root of 8 .

$$8^{\frac{1}{3}}$$

Maple keeps the result in exact form.

Maple will not automatically simplify because in many cases the root is a float. To force evaluation, use floats or try **simplify(*expr*)**:

Step 113: Find Numerical Roots

```
> simplify(8^(1/3));
```
Try to simplify $\sqrt[3]{8}$.

$$2$$

You may also enter **evalf(A)** *to produce a float.*

You may also enter **root(x,n)**:

Step 114: Use root to Find Roots

```
> root(8,3);
```
Find the cube root of 8 .

$$2$$

$2^3 = 8$.

root actually finds the ***principal root***, as explained in **?root**. Sometimes a principal root yields complex results:

Step 115: Generate Complex Root

```
> simplify((-1)^(1/3));
```
Find the cube root of -1.

$$\frac{1}{2}+\frac{1}{2}I\sqrt{3}$$

What happened to the real root, -1? $(-1)^3=-1$!

The next section demonstrates how to find real roots when the principal root is complex. See also `?sqrt` and `?roots` for related functions.

5.4.3 Real Roots

If Maple does not generate a real root and you think one exists, try a function with a rather odd name called `surd`. Just as you would use `root`, enter `surd(expr,n)`. When n is odd, then

$$\mathrm{surd}(x,n) = \begin{cases} x^{1/n} & x \geq 0 \\ -(-x)^{1/n} & x < 0 \end{cases}. \tag{5-5}$$

These equations can generate real roots, especially for odd roots of negative numbers. For instance, find the real root of $(-1)^{1/3}$:

Step 116: Real Roots

```
> surd(-1,3);
```
Find the non-complex cube root of $-1=(-1)^{1/3}$.

$$-1$$

Maple found $(-1)^{1/3}=-1$.

When no real root exists, `surd` returns a complex root. For more information, consult `?surd` and `?arithop`.

5.4.4 Symbolic Roots

You might encounter another interesting problem when taking roots of symbolic expressions. For instance, try taking the square root of x^2. You will not obtain the obvious answer x:

Step 117: Problem with Symbolic Root

```
> x:='x':
```
Unassign x.

```
> root(x^2,2);
```
Find $\sqrt{x^2}$.

$$\sqrt{x^2}$$

Maple does not know if x is positive or negative!

Maple does not know if x is positive or negative, so Maple cannot evaluate the input any further. To convince Maple that you really want x, you must tell Maple that x^2 refers to a generic symbolic value:

Step 118: Symbolic Root

```
> root(x^2,2,symbolic);
```
Find $\sqrt{x^2}$, assuming x is strictly symbolic.

$$x$$ *Maple now finds the "right" answer.*

You could also instruct Maple about properties of variables using **`assume`**. By entering **`assume(x>=0)`**, Maple "knows" that x is positive. For information on applying many kinds of properties to variables for roots and other operations, investigate **`?assume`**.

5.4.5 Logarithms

Consult Table 6-2 for a review of logarithms. A ***natural logarithm***, ln x, employs the irrational base e = 2.71828. . . . Logarithms of a general base b can be converted to ln form using the formula

$$\log_b y = \frac{\ln y}{\ln b}. \tag{5-6}$$

Maple usually expresses logarithms in terms of ln using the conversion in Eq. 5-7. For instance, find the base-10 log of 100:

Step 119: Logarithms

```
> x:=log[10](100);
```
Evaluate $\log_{10} 100 = x$, *where* $10^x = 100$.

$$x := \frac{\ln(100)}{\ln(10)}$$
`log` *tends to produce answers in terms of* $\ln$.

```
> simplify(x);
```
Also, try **`evalf`** *for floating-point values.*

$$2$$
$10^2 = 100$.

For more information about logarithms, consult **`?log`** and **`?ilog`**.

Table 5-1 Logarithms

Function	Standard Math		Maple Notation
Logarithm of Base b	$b^x = y$	$\log_b y = x$	`log[b](y)`
Base 10 Logarithm	$10^x = y$	$\log_{10} y = x$	`log[10](y)` `log10(y)`
Natural Logarithm	$e^x = y$	$\log_e y = \ln y = x$	`ln(y)` `log(y)`

5.4.6 Exponential Function

To raise the constant e to a power x, do *not* enter "**`e^x`**"! Instead enter the ***exponential function*** $\exp(x)$ which equals e^x. For instance, try the following inputs:

Step 120: Exponential Function

```
> exp(x), exp(2), exp(x)*exp(y);
```

$$e^x, e^2, e^{(xy)}$$

Enter the sequence $e^x, e^2, e^x e^y$.

Maple outputs e as $\mathbf{e}$.

Although Maple outputs `exp(expr)` as $\mathbf{e}^{expr}$, you must never enter "`e`" or "`E`" to produce the exponential function. If you wish to find e, enter `exp(1)`. Also, some new Maple users sometimes confuse the exponentiation operator caret (`^`) with the exponential function `exp`. See `?exp` for more information.

Practice!

17. Evaluate $\sqrt[3]{-8}$. Find all real and complex roots.

18. Find x such that $7^x = 163$. Show your answer as a float. Check the answer that Maple produces.

19. Evaluate the exponential constant to five decimal places.

20. Find $\ln(\exp(x))$. Discuss the relationship between the functions ln and exp.

5.5 Miscellaneous

Table 5-2 reviews common mathematical operations and functions you might encounter throughout your education and career in engineering and science. ***Procedural Maple functions***, like manipulation, evaluation, solving, plotting, and programming, are reviewed in later chapters.

Table 5-2 Miscellaneous Functions and Operations

Functions	**Standard Math**	**Maple Notation**	**Related Functions and Help**
Absolute Value	$\lvert x \rvert$	`abs(x)`	`?abs`, `?sign`, `?signum`, `?csgn`

Boolean	$x \wedge y$
	$x \vee y$
	Is $x \neq y$?
		`x and y`
		`x or y`
		`evalb(x <> y)`
			`?boolean`, `?equation`, `?evalb`, `?logic`

Complex	$\Re(x) + \Im(x)$ e^{ix}	`Re(x) + Im(x)` `exp(I*x)`	`?argument`, `?conjugate`, `?csgn`, `?evalc`, `?polar`
Factorial	$x!$	`x!` `factorial(x)`	`?binomial`, `?combinat`, `?combstruct`, `?factorial`, `?group`
Floats	$\sqrt{2} = 1.414\ldots$	`evalf(sqrt(2))`	`?evalf`, `?float`, `?fsolve`, `?numapprox`, `?trunc`
Integer	$72 = (2)^3(3)^2$	`ifactor(72)`	`?arith`, `?ifactor`, `?integer`, `?trunc`
Inverse	f^{-1}	`f@@(-1)`	`?@`, `?@@`, `?invfunc`, `readlib(invfunc)`
List	$[x_1, x_2]$ $[f(x_1), f(x_2)]$	`[x[1],x[2]]` `map(f,[x[1],x[2]])`	`?list`, `?member`, `?select`, `?sort`
Piecewise	$f(x) = \begin{cases} e^x & 0 < x \\ 0 & otherwise \end{cases}$	`piecewise(0<x,exp(x))`	`?piecewise`
Product	$\prod_{i=1}^{n} x_i$	`product(x[i],i=1..n)`	`?mul`, `?product`
Sequence	x_1, x_2, x_3	`seq(x[i],i=1..3)` `x[i] $ i=1..3`	`?seq`, `?sequence`, `?$`
Series	$\cos x = 1 - \frac{1}{2}x^2 + O(x^4)$	`series(cos(x),x=0,4)`	`?Order`, `?powseries`, `?series`, `?taylor`
Set	$x \cap y$ $x \cup y$	`x intersect y` `x union y`	`?intersect`, `?minus`, `?set`, `?union`
Summation	$\sum_{i=1}^{n} x_i$	`sum(x[i],i=1..n)`	`?add`, `?sum`

Practice!

21. Evaluate $|-18|$, $|0|$, and $|18|$.

22. Add the real and imaginary components of e^{ix}.

23. Generate the sequence $1, 2, 4, 8$ with **`seq`** or **`$`**. Assign to S the result.

24. Add each element of S. Hint: Consult **`?sum`**.

25. Multiply each element of S. Hint: Consult **`?product`**.

5.6 Functional Notation

Entering assignments in the form of **`y:=m*x+b`** provides only a shortcut for simulated functions. This section discusses how entering functions in ***functional notation*** in the form of **`f(x)`** is more natural than using simulated functions.

5.6.1 Definition

Use the operator **`->`** to create your own functions that use functional notation in the form of ***`name(args)`***. Assign the function ***`name`*** with the syntax ***`name:=args->expr`***:

- ***`name`*** defines the function name. Avoid using protected names.
- ***`args`*** are function arguments.
- ***`expr`*** is the actual expression of the function in terms of ***`args`***.

You may use zero, one, or multiple arguments in ***`args`***. Table 5-3 demonstrates the syntax for a variety of examples with different amounts of arguments. For further explanation, see the sections

Table 5-3 Functional Notation

Arguments	Standard Math	Maple Notation
0	$f() = x$	`f := () -> x`
one	$f(x) = mx + b$	`f := x -> m*x + b`
multiple	$f(x, y) = x^2 + y^2$	`f := (x,y) -> x^2 = y^2`

below and investigate **`?->`** and **`?operators[example]`**. To use a different method other than **`->`**, consult **`?unapply`**. For an alternative syntax, consult **`?student[makeproc]`**.

5.6.2 Creating Functions in Functional Notation

For instance, to create a function of one variable, like $f(x) = x^2$, use the syntax ***name := var-> expr***:

Step 121: Functional Notation with One Variable

```
> f := x -> x^2;
```

$$f := x \rightarrow x^2$$

Assign $f(x) = x^2$.

Maple now considers f as the functional form $f(x)$.

Step 122: Show Values with Functional Notation

```
> f(0),f(1),f(2),f(3);
```

$$0, 1, 4, 9$$

Use $f(x)$ to find four distinct values of x.

Do not assign x to any expression. Instead, use $f(x)$.

Practice!

26. Create a function $\mathrm{dis}(x) = x^3 + x^2 + x + 1$ using functional notation.

27. Find the value of $\mathrm{dis}(x)$ at $x = -1$ and $x = 1$.

28. Plot $\mathrm{dis}(x)$ on the interval $-1 \le 0 \le 1$.

29. Create a function $f(x, y, z) = x + y + z$ using functional notation.

5.7 Application

[To be determined]

Summary

- Functions take input, perform a task, and produce output.
- A function is a correspondence between the input (domain) and the range (output).
- You may nest Maple functions by supplying a function call as input to another function.
- You may combine functions with operators.
- Polynomials contain sums of terms with integer exponents.
- Trigonometric functions require radians for angles.
- You may find roots with **`root`** or the operator **`^`**.
- When using **`root`**, Maple finds the principal root, which may be complex.
- To find a real root, use **`surd`**.
- Maple expresses logarithms in terms of the natural log ln.
- Maple expresses the exponential function with **`exp`**.
- To create a function **`f`** with the notation **`f(x)`**, use the **`->`** operator.

Key Terms

argument
exponential function
exponentiation
factors
functional notation
Maple function
natural logarithm
nested function
parameter
polynomial
principal root
procedural Maple function
radian
root

Problems

1. What is a function? How do you express functions in Maple?
2. Given $f(x) = 2x^3$, evaluate $f(-1)$, $f(0)$, and $f(1)$ using Maple using a simulated function.
3. For Problems 3a through 3c, let $P = x^2 + 6x + 7$ and $Q = x + 1$.
   3a. Evaluate $P + Q$ and $P - Q$.
   3b. Evaluate PQ, P^2Q, and $\frac{P}{Q}$. Distribute (multiply out) all products and sums.
   3c. Divide P by Q using both **`rem`** and **`quo`**. Show the quotient and remainder in both cases. Hint: For instance, enter **`rem(P,Q,x,'q')`** for **`rem`**.
   3d. Confirm your results in Problem 3c. Hint: Try both **`evalb`** and **`expand`**.

4. Factor the polynomial $x^4 - 2x^2 + 1$. How many different roots does the polynomial contain? How many times does each root factorize the polynomial? Hint: Try `factor` and `roots`.
5. Evaluate $\sqrt[3]{-72}$. Show both real and complex roots. Hints: Use `simplify` to clarify the results. Note that Maple will show fractional components.
6. Can you take the natural logarithm of a negative number? Demonstrate your answer with a plot of $\ln x$ on $-1 \le x \le 1$.
7. Evaluate the following expressions:

7a.	$\sin^2(x)$	7f.	$\log_{10}100$
7b.	$\sin\left(\frac{\pi}{4}\right)$	7g.	$\ln 5.216$
7c.	$\sin^2\left(\frac{\pi}{4}\right)$	7h.	$2.4^{-1.2}$
7d.	$\sqrt{\sin(17)}$ (produce both exact and decimal results)	7i.	$\frac{1}{e}$ (produce both exact and decimal results)
7e.	$\tan(45°)$	7j.	πe^2 (produce both exact and decimal results)

8. Create a function that finds a trapezoid's area

$$\text{trap}(b_1, b_2, h) = h \times \left(\frac{b_1 + b_2}{2}\right)$$

using functional notation. Evaluate $\text{trap}(1, 2, 3)$ to test your function.

9. Snow blowing over large unblocked distances called *fetch* contributes to accumulating snow drifts. Given the relationship between snow transport capacity $\frac{Q_t}{Q_{inf}}$ and fetch F (m)

$$\frac{Q_t}{Q_{inf}} = \left(1 - 0.14^{\frac{F}{3000}}\right)$$

Compute F assuming $\frac{Q_t}{Q_{inf}} = 0.8$. Does transport capacity increase or decrease as fetch increases? Hints: Rearrange the equation with logarithms on both sides. Also, consider entering `assume(F>0)`.

10. [note to self - tensor transformation for angles, trig]